LETTERS TO THE EDITOR

Dear Editor:

William H. Venable has made a major contribution to the accuracy of tristimulus values calculation in his recent article.[1] The purpose of this letter is to present a basic language computer program that will give an example of Venable weights, specifically for a 10- or 20-nm measurement interval and a triangular bandpass, a shape that many monochromators approach. The ASTM tables[2] could be improved by specifying the bandpass and using Venable weights calculated with the program in the Appendix.

CIE Weights

The CIE[3] tabulates 471 zero bandpass data for illuminant and observer. Multiplying these give a set of values that define an illuminant-observer function with CIE 1-nm weights. With fluorescent illuminants, for which the CIE lists only average 5-nm bandpass values for every 5-nm interval, the illuminant data at all five wavelengths within the interval are assumed to be equal to the average. It has been suggested that fluorescent illuminant values at 1-nm intervals can be calculated from the CIE data by interpolation. This is illogical because what are needed are 1-nm bandpass data at 1-nm intervals. The interpolation of 5-nm bandpass data at 5-nm gives 5-nm bandpass data at 1-nm intervals, not 1-nm bandpass data. The difference in these two methods of handling fluorescent illuminants is probably inconsequential.

Venable Weights for Rectangular Bandpass

One contribution of Venable is the concept that a set of weights, one weight for each radiance measurement, should depend on the bandpass shape. The Venable weights are easier to understand, and the weights simpler to calculate, if the radiance measurements are made with a rectangular bandpass.

In a rectangular interval, the CIE method would calculate the segment of the tristimulus value in the interval as sum ($R \times W$) where R is the radiance of each 1-nm band in the interval and W is the weight of the corresponding illuminant-observer function. This is considered to be the correct value of the segment of the tristimulus value over this wavelength interval.

A spectrophotometer with a bandwidth of MI gives the sum of the radiances in the interval. The question is, what weight should be used to multiply the radiance sum to get the best approximation to the correct value. Venable proposes that the weight should be the sum of the CIE weights that apply to the interval. The sum of all the CIE weights that apply, in this case sum W, is the Venable weight at the wavelength of measurement. The assumption that is the basis of the Venable method is sum ($W \times R$) = sum $W \times$ sum R/MI. This assumption duplicates the correct tristimulus segment if all the W values, or all the R values, in the interval are equal.

The Venable weights calculated in this manner are of little use since few if any spectrophotometers operate with a rectangular bandpass. The calculation with the more useful triangular bandpasses follows.

TABLE I. Venable weights for specimens 1964 observer, D65 illuminant, triangular bandpass, width = MI.

NM	X	Y	Z
360	0.000	0.000	−0.001
380	−0.001	−0.001	0.007
400	0.038	0.009	0.106
420	3.012	0.286	14.073
440	7.199	1.045	36.688
460	6.544	2.508	37.613
480	1.295	4.910	14.367
500	−0.240	8.439	3.296
520	1.938	14.014	0.974
540	6.756	17.705	0.196
560	12.213	17.400	−0.017
580	16.800	14.224	0.003
600	17.779	10.129	0.000
620	13.107	5.953	0.000
640	5.894	2.413	0.000
660	1.897	0.739	0.000
680	0.464	0.181	0.000
700	0.090	0.035	0.000
720	0.021	0.008	0.000
740	0.005	0.002	0.000
760	0.001	0.000	0.000
780	0.000	0.000	0.000
800	0.000	0.000	0.000
820	0.000	0.000	0.000
SUM	94.811	100.000	107.305

Venable Weights for Triangular Bandpasses

When a single weight multiplies a radiance measurement made with a symmetrical triangular bandpass, the proportion of the weight that is applied to the measurement interval centered at the measurement wavelength is 3/4 of the weight. Also, 1/8 of the weight is applied to each adjacent interval. The total weight that is applied to an interval is 3/4 the weight at the measurement wavelength plus 1/8 of the two adjacent weights. Venable proposes that the sum of these applied weights equal the sum of the CIE weights in the interval. Expressed as an equation,

$$1/8\ W(i-Mi) + 3/4\ W(i) + 1/8\ W(i+MI) = A(i) \quad (1)$$

where W is the Venable weight

CCC 0361-2317/90/020117-03$4.00

A is the sum of the CIE weights over the interval.

Equation (1) is useful in checking a set of weights. 1/8 of one weight, plus 3/4 of the next, plus 1/8 of the next should always equal the sum of the CIE weights at the middle interval.

To calculate each $W(i)$, the following equation may be used[4].

$$W(i) = \mathrm{sqr}(2)^* \sum_{k=360}^{820} (2^*\mathrm{sqr}(2)-3)^{\mathrm{abs}(k-i)} {}^*A(k) \quad (2)$$

The algorithm to solve equation (2) is best understood by reading the basic language program in the Appendix.

To calculate weights for light sources make all values of the illuminant 1 and proceed as above.

Table I gives the Venable weights for the 1964 observer and illuminant D65. The correctness of Table I may be checked by observing that the sum of 1/8, 3/4, and 1/8 of successive weights equals the sum of the 1-nm CIE weights in the central interval. Table I may be used to check any computer program that one may write.

Appendix

```
100 REM E.I. STEARNS 10/5/89
110 REM CALCULATE VENABLE WEIGHTS FOR
    TRIANGULAR BANDPASS
120 REM AT START, W=CIE 1-NM WEIGHTS
130 REM AT END, AT MEASUREMENT
    WAVELENGTHS, W=VENABLE WEIGHTS
140 REM A=SUM OF CIE WEIGHTS IN
    MEASUREMENT INTERVAL
150 DIM A(470,2),W(470,2)
160 REM ***
170 REM PUT 1-NM CIE WEIGHTS IN W
180 INPUT "MEASUREMENT INTERVAL
    10/20 ";MI
190 FOR J = 0 TO 2
200 GOSUB 260: REM FILL STANDARD SUMS A
210 GOSUB 490: REM CALCULATE VENABLE
    WEIGHTS
220 NEXT J
230 REM PRINT VENABLE WEIGHTS
240 END
250 REM ***
260 REM FILL A WITH STD CIE WEIGHT SUMS
270 FOR K = 0 + MI TO 460 - MI STEP MI
280 A(K,J) = 0
290 FOR M = - MI / 2 + 1 TO MI / 2 - 1
300 A(K,J) = A(K,J) + W(K + M,J)
310 NEXT M
320 A(K,J) = A(K,J) + .5 * W(K - MI / 2,J) + .5
    * W(K + MI / 2,J)
330 NEXT K
340 A(0,J) = A(0,J) + .5 * W(MI / 2,J)
350 A(460,J) = A(460,J) + .5 * W(460 - MI / 2,J)
360 FOR M = 0 TO MI / 2 - 1
370 A(0,J) = A(0,J) + W(M,J)
380 A(460,J) = A(460,J) + W(460 - M,J)
390 NEXT M
400 GOSUB 430
410 RETURN
420 REM ***
430 REM ZERO INITIAL W
440 FOR K = 0 TO 460 STEP MI
450 W(K,J) = 0
460 NEXT K
470 RETURN
480 REM ***
490 REM CALCULATE VENABLE WEIGHTS
500 REM SUBROUTINE WRITTEN BY RICHARD
    E. STEARNS
510 REM WANT W(I,J) SUCH THAT
    (W(I-MI,J)+6*W(I,J)+W(I+MI))/8 = A(I,J)
520 DD = SQR (2) * 2 - 3
530 REM THE W(I,J) ARE EXPRESSED IN
    TERMS OF THE A(I,J) BY THE FOLLOWING:
540 REM W(I,J) = ( L(I) + A(I,J) + U(I) ) *
    SQR(2)
550 REM WHERE L(I) = SUM FOR K≧1 OF
    A(I-MI*K,J)*DD^K
560 REM WHERE U(I) = SUM FOR K≧ 1 OF
    A(I+MI*K,J)*DD^K
570 REM THE L(I) ARE EASILY COMPUTED IN
    LEFT TO RIGHT ORDER USING:
580 REM L(I+MI) = DD*(A(I,J)+L(I))
590 REM THE FIRST LOOP COMPUTES THE L(I)
    AND SETS W(I,J)=L(I)+A(I,J)
600 SD = 0
610 REM SINCE A(I,J)=0 FOR I<0, SD=0=L(0)
620 FOR I = 0 to 460 STEP MI
630 REM HERE SD=L(I)
640 SD = SD + A(I,J)
650 REM SD=L(I)+A(I,J)
660 W(I,J) = SD
670 REM W(I,J) IS NOW L(I)+A(I,J)
680 SD = SD * DD
690 REM SD=DD*(L(I)+A(I,J))=L(I+MI)
700 NEXT I
710 REM THE U(I) ARE EASILY COMPUTED IN
    RIGHT TO LEFT ORDER USING:
720 REM U(I-MI) = DD*(A(I,J)+U(I))
730 REM THE NEXT LOOP COMPUTES THE U(I)
    AND ADDS THEM TO THE W(I,J)
740 REM THE W(I,J) BECOME L(I)+A(I,J)+U(I)
750 SD = 0
760 REM SINCE A(I,J)=0 FOR I>460,
    SD=0=U(460)
770 FOR I = 460 TO 0 STEP - MI
780 REM HERE SD=U(I)
790 W(I,J) = W(I,J) + SD
800 REM NOW A(I,J) = L(I)+A(I,J)+U(I)
810 SD = DD * (SD +A(I,J))
820 REM SD = DD*(U(I)+A(I,J)) = U(I-MI)
```

```
830  NEXT I
840  REM  NEXT MULTIPLY THE W(I,J) BY
     SQR(2) TO GET DESIRED VALUES
850  SR = SQR (2)
860  FOR I = 0 TO 460 STEP MI
870  W(I,J) = W(I,J) * SR
880  NEXT I
890  RETURN
```

E. I. STEARNS
321 Woodland Way
Clemson, SC 29631

1. William H. Venable, *Color Res. Appl.* **14**, 260–267 (1989).
2. ASTM E308-85 Standard method for computing the color of objects using the CIE system. (available from ASTM 1916 Race St., Philadelphia, PA 19103).
3. Publication CIE No.15.2, (TC-1,3), *Colorimetry*, 2nd ed., 1984, "Official Recommendations of the International Commission on Illumination", Bureau Central de la CIE, Paris.
4. Richard E. Stearns, private communication.

Reply:

As a condition for optimizing the weights, Dr. Stearns requires that the integral of $D(\lambda)$ be exactly zero over each wavelength range equal to the measurement interval and centered on the nominal wavelength of measurement. In this very reasonable approach to my second criterion for optimum weights, the interval over which $D(\lambda)$ integrates to zero need not be centered at any particular wavelength. Instead, for example, in a set of measurements at 20-nm intervals, one might require that the integral of the defining CIE weight function from 480 to 500 nm be equal to the sum of 1/2 of the optimum weight at 480 nm plus 1/2 the optimum weight at 500 nm. This would yield a set of weights that differ very slightly from those in Table I of Dr. Stearns' letter and which could be averaged with them to produce an "improved" optimization in terms of a more uniform expectation of error. In principle, an even better set of optimum weights could be obtained by averaging twenty such sets constructed on twenty intervals each displayed 1 nm from its neighbors. When the passband set is far from optimum, minimizing the integral of the square of $D(\lambda)$ over the full wavelength range subject to keeping the correct sum of weights provides the minimum expectation of error. However, when applied to an optimum set of passbands, such as triangular passbands at intervals equal to the half-height bandwidth, weights determined on a single set of intervals as in Dr. Stearns' program differ negligible from those determined by more complex schemes.

WILLIAM H. VENABLE
Hunter Associates Laboratory, Inc.
11491 Sunset Hills Rd.
Reston, VA 22090